Vektoren, Geraden ,Ebenen

Prof. Dr. Dörte Haftendorn, MuPAD 4, http://haftendorn.uni-lueneburg.de Aug.06

Automatische Übersetzung aus MuPAD 3.11, Mrz 06 Update 14.03.06

Es fehlen nocht textliche Änderungen, die MuPAD 4 direkt berücksichtigen, das ist in Arbeit.

Web: http://haftendorn.uni-lueneburg.de www.mathematik-verstehen.de

+++

1.Vektoren 2D

Vektoren werden als Matrizen aufgefasst.
Eingabe einer "flachen" Liste erzeugt Spaltenvektoren.

```
a := matrix([3,2]);
v := matrix([6,-2]);
```

$$\begin{pmatrix} 3 \\ 2 \end{pmatrix}$$

$$\begin{pmatrix} 6 \\ -2 \end{pmatrix}$$

Definition eines Zeilenvektors mit Doppelliste:

```
az := matrix([[3,2]]);
```

$$\begin{pmatrix} 3 & 2 \end{pmatrix}$$

Hier wird eine Gerade als Funktion ihres Parameters definiert.
Es ginge aber auch als Term g:=a+r*v

```
gv:=r->a+r*v
```

$$r \rightarrow a + r \cdot v$$

```
gv(r)
```

$$\begin{pmatrix} 6 \cdot r + 3 \\ 2 - 2 \cdot r \end{pmatrix}$$

```
p1:=gv(1)
```

$$\begin{pmatrix} 9 \\ 0 \end{pmatrix}$$

Zum Zeichnen nimmt man den Pfeil-Befehl.
Es werden drei Graphik-Primitive erzeugt, die dann von plot dargestellt werden.

```
ag:=plot::Arrow2d(a);
p1g:=plot::Arrow2d(a+v);
vg:=plot::Arrow2d(a,a+v);
```

plot::Arrow2d$([0,\ 0],\ [3,\ 2])$

plot::Arrow2d$([0,\ 0],\ [9,\ 0])$

plot::Arrow2d$([3,\ 2],\ [9,\ 0])$

```
plot(ag,vg,p1g)
```



Eindrucksvolle annimierte Geradendarstellung:

```
vani:=plot::Arrow2d(a,gv(r),r=-2..2)
```

plot::Arrow2d$([3,\ 2],\ [6 \cdot r + 3,\ -2 \cdot r + 2])$

```
plot(ag,vani)
```



Zum Animieren: Doppeklick in der Graphik, dann oben Player bedienen.

Herausgreifen der Zeilen aus den Verktoren:

```
gv(r)[1]
```

$$6 \cdot r + 3$$

Herstellen der üblichen 2d-Geradengleichung

```
gr:=solve({gv(r)[1]=x,gv(r)[2]=y},{y,r})
```

$$\left\{\left[r = \frac{x}{6} - \frac{1}{2}, y = 3 - \frac{x}{3}\right]\right\}$$

Hinten kann mann die Gleichung ablesen.

Automatisches Herausgreifen:

```
gerade:=gr[1][2][2]
```

$$3 - \frac{x}{3}$$

```
geradeg:=plot::Function2d(gerade,x=-10..16, LineColor=RGB::Red):
plot(geradeg,ag,vani);
```



Zum Animieren: Doppeklick in der Graphik, dann oben Player bedienen.

2.Vektoren 3D ###

Vektoren werden als Matrizen aufgefasst.

Eingabe einer "flachen" Liste erzeugt Spaltenvektoren.

```
a := matrix([3,2,-1]);
v := matrix([6,-2,2]);    o:=matrix([0,0,0]):
```

$$\begin{pmatrix} 3 \\ 2 \\ -1 \end{pmatrix}$$

$$\begin{pmatrix} 6 \\ -2 \\ 2 \end{pmatrix}$$

Hier wird eine Gerade als Funktion ihres Parameters definiert.

Es ginge aber auch als Term g:=a+r*v

```
gv:=r->a+r*v
```

$$r \rightarrow a + r \cdot v$$

```
gv(r); gv(1)
```

$$\begin{pmatrix} 6 \cdot r + 3 \\ 2 - 2 \cdot r \\ 2 \cdot r - 1 \end{pmatrix}$$

$$\begin{pmatrix} 9 \\ 0 \\ 1 \end{pmatrix}$$

Zum Zeichnen nimmt man den Pfeil-Befehl.
Es werden drei Graphik-Primitive erzeugt, die dann von plot dargestellt werden

```
ag:=plot::Arrow3d(a);
p1g:=plot::Arrow3d(a+v);
vag:=plot::Arrow3d(a,a+v);
```

plot::Arrow3d([0, 0, 0], [3, 2, − 1])

plot::Arrow3d([0, 0, 0], [9, 0, 1])

plot::Arrow3d([3, 2, − 1], [9, 0, 1])

```
plot(ag,vag,p1g)
```



3D-Graphik "anfassen", d.h. Doppelkick in der Graphik, dann mit Maus drehen.
Verändern der Eigenschaften:

```
ag::LineColor:=RGB::Green:
vag::LineColor:=RGB::Magenta:
p1g::LineStyle:=Dashed:
```

```
plot(ag,vag,p1g, Axes=Origin)
```



Funktion, die 3 Vektoren in eine Liste umwandelt

```
vek3Liste:=(v1,v2,v3)->[[ v1[i]$i=1..3 ],[ v2[i]$i=1..3 ],[ v3[i]$i=
```

```
vek3Liste(o,a,a+v)
```

$$[[0, 0, 0], [3, 2, -1], [9, 0, 1]]$$

3d-Polygone können genau dann gefüllt werden, wenn es sich um Dreiecke handelt.

```
dreieck:=plot::Polygon3d(vek3Liste(o,a,a+v),Filled=TRUE)
```

plot::Polygon3d([[0, 0, 0], [3, 2, − 1], [9, 0, 1]])

```
plot(dreieck,ag,vag,plg, Axes=Origin)
```



Beim Drehen der Graphik kann man nun schön die räumliche Anordnung sehen.

3. Zwei Geraden ##

```
w:=matrix([1,1,3]);
wag:=plot::Arrow3d(a,a+w, LineColor=RGB::Black);
```

$$\begin{pmatrix} 1 \\ 1 \\ 3 \end{pmatrix}$$

plot::Arrow3d([**3, 2, − 1**], [**4, 3, 2**])

```
b:=2*a;
bg:=plot::Arrow3d(b)
```

$$\begin{pmatrix} 6 \\ 4 \\ -2 \end{pmatrix}$$

plot::Arrow3d([**0, 0, 0**], [**6, 4, − 2**])

```
gw:=r->b+r*w
```

$$r \to b + r \cdot w$$

```
solve(gw(r)=gv(s),{r,s})
```

$$\emptyset$$

Das überrascht nicht. Die Geraden sind windschief.

Zeichnen von Geraden im Raum

Die einfachste Art ist, den Orts- und den Richtungsvektor (oder ein Vielfaches) zu zeichnen

```
gvg:=plot::Arrow3d(a-2*v,a+2*v):
gwg:=plot::Arrow3d(b-2*w,b+2*w):
 plot(ag,gvg,bg,gwg, Axes=Origin)
```



4. Ebenen

Nun soll eine Ebene definiert werden, diesmal ohne Funktion.

```
eb:=a+r*v+s*w:
[a,v,w,eb];
```

$$\left[\begin{pmatrix}3\\2\\-1\end{pmatrix},\begin{pmatrix}6\\-2\\2\end{pmatrix},\begin{pmatrix}1\\1\\3\end{pmatrix},\begin{pmatrix}6\cdot r+s+3\\s-2\cdot r+2\\2\cdot r+3\cdot s-1\end{pmatrix}\right]$$

Die Angabe als Liste ist eine Lesehilfe.

```
plot(ag,vag,wag)
```



Bessere Sicht mit Ebenendreieck

```
ebene:=plot::Polygon3d(vek3Liste(a,a+v,a+w),Filled=TRUE);
```

plot::Polygon3d([[3, 2, − 1], [9, 0, 1], [4, 3, 2]])

```
[a,v,w]
```

$$\left[\begin{pmatrix}3\\2\\-1\end{pmatrix}, \begin{pmatrix}6\\-2\\2\end{pmatrix}, \begin{pmatrix}1\\1\\3\end{pmatrix}\right]$$

```
[a,a+v,a+w]
```

$$\left[\begin{pmatrix}3\\2\\-1\end{pmatrix}, \begin{pmatrix}9\\0\\1\end{pmatrix}, \begin{pmatrix}4\\3\\2\end{pmatrix}\right]$$

```
ag;vag;wag;
```

plot::Arrow3d([0, 0, 0], [3, 2, − 1])

plot::Arrow3d([3, 2, − 1], [9, 0, 1])

plot::Arrow3d([3, 2, − 1], [4, 3, 2])

```
plot(ebene,ag,vag,wag)
```



5. Gerade und Ebene

```
c:=matrix([-1,2,3]);
cg:=plot::Arrow3d(c);
u:=matrix([8,0,-4]);
ucg:=plot::Arrow3d(c,c+u,LineColor=RGB::Red);
```

$$\begin{pmatrix} -1 \\ 2 \\ 3 \end{pmatrix}$$

plot::Arrow3d([0, 0, 0], [−1, 2, 3])

$$\begin{pmatrix} 8 \\ 0 \\ -4 \end{pmatrix}$$

plot::Arrow3d([−1, 2, 3], [7, 2, −1])

```
plot(cg,ucg,ebene,ag,vag,wag)
```




```
gu:=r->c+r*u;
```

$$r \rightarrow c + r \cdot u$$

```
eb=gu(t)
```

$$\begin{pmatrix} 6 \cdot r + s + 3 \\ s - 2 \cdot r + 2 \\ 2 \cdot r + 3 \cdot s - 1 \end{pmatrix} = \begin{pmatrix} 8 \cdot t - 1 \\ 2 \\ 3 - 4 \cdot t \end{pmatrix}$$

```
solve(eb=gu(t),{r,s,t})
```

$$\left\{\left[r = \frac{1}{6}, s = \frac{1}{3}, t = \frac{2}{3}\right]\right\}$$

Die Ebene und die Gerade schneiden sich.

Probe und Schnittpunkt

```
[subs(eb, r=1/6, s= 1/3), gu(2/3)]
```

$$\left[\begin{pmatrix}\frac{13}{3}\\2\\\frac{1}{3}\end{pmatrix},\begin{pmatrix}\frac{13}{3}\\2\\\frac{1}{3}\end{pmatrix}\right]$$

6. Mehrere Ebenen

```
ebe:=(r,s)->c+r*u+s*(v+w): [c,u,v+w]; ebe
```

$$\left[\begin{pmatrix}-1\\2\\3\end{pmatrix},\begin{pmatrix}8\\0\\-4\end{pmatrix},\begin{pmatrix}7\\-1\\5\end{pmatrix}\right]$$

$$(r, s) \to c + r \cdot u + s \cdot (v + w)$$

```
ebeg:=plot::Polygon3d(vek3Liste(c,c+u,c+(v+w)),
Filled=TRUE, FillColor=[1,0.5,0.5]);plot(ebeg)
```

plot::Polygon3d([[− 1, 2, 3], [7, 2, − 1], [6, 1, 8]])




```
plot(cg,ucg,ebeg,ag,vag,wag,ebene)
```




```
eb=ebe(t,k)
```

$$\begin{pmatrix} 6\cdot r+s+3 \\ s-2\cdot r+2 \\ 2\cdot r+3\cdot s-1 \end{pmatrix} = \begin{pmatrix} 7\cdot k+8\cdot t-1 \\ 2-k \\ 5\cdot k-4\cdot t+3 \end{pmatrix}$$

```
solve(eb=ebe(t,k),{t,k,r})
```

$$\left\{\left[k = s-\frac{1}{3},\ r = s-\frac{1}{6},\ t = \frac{2}{3}\right]\right\}$$

Bei drei Gleichungen für die 4 Parameter ist zu erwarten,
dass i.a. ein Parameter unbestimmt bleibt. Hier ist es das s.
Man kann also ablesen, dass Die folgende Gerade Schnittgerade ist:
Übrigens hier zeigt sich, dass es günstig ist, Geraden und Ebenen als Funktionen ihrer Parameter aufzufassen.

```
gs:=s->ebe(2/3,s-1/3);
gsg:=plot::Arrow3d(gs(0),gs(1))
```

$$s \to \mathrm{ebe}\left(\frac{2}{3},\ s-\frac{1}{3}\right)$$

$$\mathrm{plot::Arrow3d}\left(\left[2,\ \frac{7}{3},\ -\frac{4}{3}\right],\ \left[9,\ \frac{4}{3},\ \frac{11}{3}\right]\right)$$

```
plot(cg,ucg,ebeg,ag,vag,wag,ebene,gsg)
```



Orts und Richtungsvektor der Schnittgeraden sind:
Oben ist nur der Richtungsvektor passend eingezeichnet.

```
[gs(0),gs(1)-gs(0)]
```

$$\left[\begin{pmatrix} 2 \\ \frac{7}{3} \\ -\frac{4}{3} \end{pmatrix}, \begin{pmatrix} 7 \\ -1 \\ 5 \end{pmatrix}\right]$$

7. Skalarprodukt, Hessesche Normalform.... --> Extraseite

```
[a,c,a.c]
```

$$\left[\begin{pmatrix} 3 \\ 2 \\ -1 \end{pmatrix}, \begin{pmatrix} -1 \\ 2 \\ 3 \end{pmatrix}, \begin{pmatrix} 3 & -1 \\ 2 & 2 \\ -1 & 3 \end{pmatrix}\right]$$

```
_plus(x[i]*y[i] $i=1..3);
```

$$x_1 \cdot y_1 + x_2 \cdot y_2 + x_3 \cdot y_3$$

```
skalar3:=(x,y)-> _plus(x[i]*y[i] $i=1..3):
[a,c,skalar3(a,c)]
```

$$\left[\begin{pmatrix} 3 \\ 2 \\ -1 \end{pmatrix}, \begin{pmatrix} -1 \\ 2 \\ 3 \end{pmatrix}, -2\right]$$